```
>
```

Eigenvalues and Eigenvectors

```
> restart;
> with(LinearAlgebra):
```

We illustrate the process of finding **eigenvalues** and **eigenvectors** with a 2x2 matrix.

```
> A:=Matrix(2,2,[[1,1],[4,1]]);
```

$$A := \begin{bmatrix} 1 & 1 \\ 4 & 1 \end{bmatrix}$$

We will need to use the 2x2 identity matirx.

```
> Id:=IdentityMatrix(2);
```

$$Id := \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

We find the eigenvalues by solving the characteristic polynomial. We first find the characteristic polynomial from the definition.

```
> p:=Determinant(A-lambda*Id);
```

$$p := -3 - 2\,\lambda + \lambda^2$$

We get the same or opposite polynomial from CharacteristicPolynomial since it uses $\det(\lambda\, I - A)$ instead of $\det(A - \lambda\, I)$. However, each has the same roots or zeroes.

```
> p:=CharacteristicPolynomial(A,lambda);
```

$$p := -3 - 2\,\lambda + \lambda^2$$

We solve the characteristic equation to find the eigenvalues, which may be real or complex.

```
> eigval:=solve(p=0,lambda);
```

$$eigval := 3, -1$$

For each eigenvalue, we find corresponding eigenvectors. Note that the scalar multiple of an eigenvector is also an eigenvector corresponding to the same eigenvalue. For each eigenvalue λ, we solve the equation (A-λI)***x**=**0** to find corresponding eigenvectors.

```
> z:=Matrix(2,1);
  for i from 1 to 2 do
  eigvec[i]:=LinearSolve(<A-eigval[i]*Id|z>);
  od;
```

$$z := \begin{bmatrix} 0 \\ 0 \end{bmatrix}$$

$$eigvec_1 := \begin{bmatrix} \_t_1 \\ 2\,\_t_1 \end{bmatrix}$$

$$eigvec_2 := \begin{bmatrix} \_t0_1 \\ -2\,\_t0_1 \end{bmatrix}$$

We can find specific eigenvectors by substituting any real or complex number for $\_t0_i$, for example 1.

A second way to find eigenvalues and eigenvectors is to use the commands Eigenvalues and Eigenvectors from the **LinearAlgebra** package to find the eigenvalues and eigenvectors.

```
> eigval:=Eigenvalues(A,output=list);
```

$$eigval := [3, -1]$$

```
> v:=Eigenvectors(A,output=list);
```

$$v := \left[\left[-1, 1, \left\{\left[\begin{array}{c} -\frac{1}{2} \\ 1 \end{array}\right]\right\}\right], \left[3, 1, \left\{\left[\begin{array}{c} \frac{1}{2} \\ 1 \end{array}\right]\right\}\right]\right]$$

The following command picks out the second eigenvector along with its eigenvalue from the list.

```
> v[2];
```

$$\left[3, 1, \left\{\left[\begin{array}{c} \frac{1}{2} \\ 1 \end{array}\right]\right\}\right]$$

The first entry gives the eigenvalue corresponding to the eigenvector, and the second its multiplicity.

```
> v[2,1];
```

$$3$$

```
> v[2,2];
```

$$1$$

The third entry is a list of basis eigenvectors for the particular eigenvalue. Eigenvalues of multiplicity greater than 1 may have multiple basis eigenvectors. You can get all the eigenvectors corresponding to the particular eigenvalue by taking all linear combinations of these eigenvectors.

```
> v[2,3];
```

$$\left\{\left[\begin{array}{c} \frac{1}{2} \\ 1 \end{array}\right]\right\}$$

The next command picks out the first (and in this case only) basis eigenvector.

```
> v[2,3,1];
```

$$\left[\begin{array}{c} \frac{1}{2} \\ 1 \end{array}\right]$$

Finally, we pick out the second component of this eigenvector.

```
> v[2,3,1][2];
```

$$1$$

Next, we look at the case where we have complex eigenvalues.

```
> A:=Matrix(2,2,[[2,4],[-4,2]]);
```

$$A := \left[\begin{array}{cc} 2 & 4 \\ -4 & 2 \end{array}\right]$$

```
> p:=CharacteristicPolynomial(A,lambda);
```

$$p := \lambda^2 - 4\lambda + 20$$

```
> eigval:=solve(p,lambda);
```

$$eigval := 2 + 4\,I,\ 2 - 4\,I$$

```
> for i from 1 to 2 do
  eigvec[i]:=LinearSolve(<A-eigval[i]*Id|z>);
  od;
```

$$eigvec_1 := \begin{bmatrix} -I\,\_t1_2 \\ \_t1_2 \end{bmatrix}$$

$$eigvec_2 := \begin{bmatrix} I\,\_t2_2 \\ \_t2_2 \end{bmatrix}$$

We notice that complex eigenvalues come in conjugate pairs.

```
> eigval:=Eigenvalues(A,output=list);
```

$$eigval := [2 + 4\,I,\ 2 - 4\,I]$$

```
> v:=Eigenvectors(A,output=list);
```

$$v := \left[\left[2 + 4\,I,\ 1,\ \left\{\begin{bmatrix} -I \\ 1 \end{bmatrix}\right\}\right],\ \left[2 - 4\,I,\ 1,\ \left\{\begin{bmatrix} I \\ 1 \end{bmatrix}\right\}\right]\right]$$

We nowlook at an example of a matrix with an eigenvalue of multiplicity greater than 1.

```
> A:=Matrix(2,2,[[-2,0],[0,-2]]);
```

$$A := \begin{bmatrix} -2 & 0 \\ 0 & -2 \end{bmatrix}$$

```
> Eigenvalues(A,output=list);
```

$$[-2, -2]$$

```
> Eigenvectors(A,output=list);
```

$$\left[\left[-2,\ 2,\ \left\{\begin{bmatrix} 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \end{bmatrix}\right\}\right]\right]$$

We are told that the eigenspace for $\lambda = -2$ consists of all vectors that are linear combinations of $\begin{bmatrix} 1 \\ 0 \end{bmatrix}$ and $\begin{bmatrix} 0 \\ 1 \end{bmatrix}$.

```
>
>
```